AULA 3
Software for Algebra and Geometry Experimentation - SAGE

Differentiation:

The limit of this ratio when $\Delta\, x$ approaches the limit zero is, from our definition, the derivative and is denoted by the symbol $\frac{dy}{dx}$. Therefore

$$\frac{dy}{dx} = \lim_{\Delta x \to 0} \frac{f(x+\Delta x) - f(x)}{\Delta x}.$$

defines the *derivative of y [or $f(x)$] with respect to* x. From (3.3), we also get

$$\frac{dy}{dx} = \lim_{\Delta x \to 0} \frac{\Delta y}{\Delta x}$$

The process of finding the derivative of a function is called *differentiation*.

O cálculo diferencial e seus conceitos principais.

Exercises:

```
sage: x = var("x")
sage: h = var("h")
sage: f(x) = 3*x^2 + 5
sage: Deltay = f(x+h)-f(x)
sage: Deltay
3*(h + x)^2 - 3*x^2
sage: (Deltay/h)
3*((h + x)^2 - x^2)/h
sage: (Deltay/h).expand()
3*h + 6*x
sage: limit((f(x+h)-f(x))/h,h=0)
6*x
sage: diff(f(x),x)
6*x
```

```
sage: y = 3/(x^2-1)
sage: y
3/(x^2 - 1)
sage: diff(y,x)
-6*x/(x^2 - 1)^2
```

O cálculo diferencial e seus conceitos principais.

```
sage: y = 3/(x^2-1)
sage: y
3/(x^2 - 1)
sage: diff(y,x)
-6*x/(x^2 - 1)^2
sage: t = var("t")
sage: diff(acos(t),t)
-1/sqrt(-t^2 + 1)
sage: v = var("v")
sage: diff(acsc(v),v)
-1/(sqrt(v^2 - 1)*v)
sage: x = var("x")
sage: u = function('u',x)
sage: u
u(x)
sage: v = function('v',x)
sage: v
v(x)
sage: diff(u*v,x)
v(x)*diff(u(x), x) + u(x)*diff(v(x), x)
sage: diff(v*u,x)
v(x)*diff(u(x), x) + u(x)*diff(v(x), x)
sage: diff(u/v,x)
diff(u(x), x)/v(x) - u(x)*diff(v(x), x)/v(x)^2
sage: diff(v/u,x)
-v(x)*diff(u(x), x)/u(x)^2 + diff(v(x), x)/u(x)
```

$$\frac{d\arccos t}{dt} = -\frac{1}{\sqrt{1-t^2}} \text{ and } \frac{d\operatorname{arccsc} v}{dv} = -\frac{1}{v\sqrt{v^2-1}}.$$

$$\frac{d}{dx}(uv) = u\frac{dv}{dx} + v\frac{du}{dx}$$

$$\frac{d}{dx}\left(\frac{u}{v}\right) = \frac{v\frac{du}{dx} - u\frac{dv}{dx}}{v^2}$$

```
sage: diff(sin(v),x)
cos(v(x))*diff(v(x), x)
sage: diff(arcsin(v),x)
diff(v(x), x)/sqrt(-v(x)^2 + 1)
```

$$\frac{d}{dx}(\sin\ v) = \cos\ v\frac{dv}{dx}$$

$$\frac{d}{dx}(\arcsin\ v) = \frac{\frac{dv}{dx}}{\sqrt{1-v^2}}$$

O cálculo diferencial e seus conceitos principais.

This simply computes $\frac{d}{dt}(e^{2t}\cos(t)$ in two ways (one: directly, the second: using the product rule) and checks that they are the same.

```
sage: t = var("t")
sage: f = cos(t)
sage: g = exp(2*t)
sage: diff(f*g,t)
2*cos(t)*e^(2*t) - e^(2*t)*sin(t)
sage: diff(f,t)*g+f*diff(g,t)
2*cos(t)*e^(2*t) - e^(2*t)*sin(t)
```

$$\frac{\frac{d}{dx}(uv)}{uv} = \frac{\frac{du}{dx}}{u} + \frac{\frac{dv}{dx}}{v}.$$

Produto de um número finito de funções:

$$\begin{aligned}
\frac{\frac{d}{dx}(v_1 v_2 \cdots v_n)}{v_1 v_2 \cdots v_n} &= \frac{\frac{dv_1}{dx}}{v_1} + \frac{\frac{d}{dx}(v_2 v_3 \cdots v_n)}{v_2 v_3 \cdots v_n} \\
&= \frac{\frac{dv_1}{dx}}{v_1} + \frac{\frac{dv_2}{dx}}{v_2} + \frac{\frac{d}{dx}(v_3 v_4 \cdots v_n)}{v_3 v_4 \cdots v_n} \\
&= \frac{\frac{dv_1}{dx}}{v_1} + \frac{\frac{dv_2}{dx}}{v_2} + \frac{\frac{dv_3}{dx}}{v_3} + \cdots + \frac{\frac{dv_n}{dx}}{v_n} \frac{d}{dx}(v_1 v_2 \cdots v_n) \\
&= (v_2 v_3 \cdots v_n)\frac{dv_1}{dx} + (v_1 v_3 \cdots v_n)\frac{dv_2}{dx} + \cdots + (v_1 v_2 \cdots v_{n-1})\frac{dv_n}{dx}.
\end{aligned}$$

```
sage: f = function('f', t)
sage: g = function('g', t)
sage: (f(t)*g(t)).diff(t) # product rule for 2 functions
g(t)*diff(f(t), t) + f(t)*diff(g(t), t)
sage: h = function('h', t)
sage: (f(t)*g(t)*h(t)).diff(t) # product rule for 3 functions
g(t)*h(t)*diff(f(t), t) + f(t)*h(t)*diff(g(t), t) + f(t)*g(t)*diff(h(t), t)
```

O cálculo diferencial e seus conceitos principais.

Diferenciação de funções compostas:

For example, if $y = \frac{2v}{1-v^2}$, and $v = 1 - x^2$

```
sage: t = var('t')
sage: f = function('f', t)
sage: g = lambda v: 2*v/(1-v^2)
sage: g
<function <lambda> at 0x6fffda858c0>
sage: v
v(x)
sage: g(f(t)).diff(t) # this gives the general form, for any f
4*f(t)^2*diff(f(t), t)/(f(t)^2 - 1)^2 - 2*diff(f(t), t)/(f(t)^2 - 1)
sage: f = lambda x: 1-x^2
sage: f
<function <lambda> at 0x6fecd0c00c8>
sage: g(f(t)).diff(t) # this gives the specific answer in this case
-8*(t^2 - 1)^2*t/((t^2 - 1)^2 - 1)^2 + 4*t/((t^2 - 1)^2 - 1)
```

Diferenciação de funções inversas:

$$f'(x) = \frac{1}{\phi'(y)}.$$

- $y = x^2 + 1,\ x = \pm\sqrt{y-1}.$
- $y = a^x,\ x = \log_a y.$
- $y = \sin\ x,\ x = \arcsin\ y.$

Recall that a function $g(x)$ is the inverse of a given function $f(x)$ if $f(g(x)) = g(f(x)) = x$.

Diferenciação de funções inversas:

- $y = x^2 + 1, x = \pm\sqrt{y-1}.$

```
sage: f(x) = x^2+1
sage:  g= plot (f(x),x ,-1,2, figsize =3)
sage: g
Launched png viewer for Graphics object consisting
of 1 graphics primitive
sage: g= plot (f(x),x ,0 ,5, figsize =3)
sage: g
Launched png viewer for Graphics object consisting
of 1 graphics primitive
sage: var('x,y')
(x, y)
sage: sol= solve (f(y)==x,y)
sage: sol
[y == -sqrt(x - 1), y == sqrt(x - 1)]
sage: g(x)= sol [0]. rhs ()
sage: g(x)
-sqrt(x - 1)
sage: g(x)= sol [1]. rhs ()
sage: g(x)
sqrt(x - 1)
sage: g(f(x)). simplify_full ()
sqrt(x^2)
sage: f(x)
x^2 + 1
sage: g(x)
sqrt(x - 1)
sage: f(g(x)). simplify_full ()
x
sage: h= plot ((f(x),g(x)),x ,1 ,5 , figsize =3,
ymin =1, ymax =5)
```







Diferenciação de funções inversas:

The derivative of the inverse function is equal to the reciprocal of the derivative of the direct function.

- $y = x^2 + 1, x = \pm\sqrt{y-1}.$

```
sage: var('f,g,x,y,fi')
(f, g, x, y, fi)
sage: f(x) = x^2+1
sage: fi= solve (f(y)==x,y)
sage: fi
[y == -sqrt(x - 1), y == sqrt(x - 1)]
sage: g(x) = fi[1] .rhs()
sage: g(x)
sqrt(x - 1)
sage: diff(g(x))
1/2/sqrt(x - 1)
sage: diff(f(x))
2*x
sage: 1/diff(g(x))
2*sqrt(x - 1)

sage: sol(x)=diff(f(x))
sage: sol(x)
2*x
sage: fi= solve (sol(y)==x,y)
sage: fi
[y == 1/2*x]
```

$$\frac{dy}{dx} = \frac{1}{\frac{dx}{dy}},$$

$$f'(x) = \frac{1}{\phi'(y)}.$$

Diferenciação de funções inversas:

Determine if the function $f(x) = 2x^3 + 3x$ has an inverse.

```
sage: var('f,g,x,y, sol')
(f, g, x, y, sol)
sage: f(x) = 2*x^3 + 3*x
sage: sol= solve (f(y)==x,y)
sage: sol
[
y == -1/2*(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3)*(I*sqrt(3) + 1) + 1/4*(-I*sqrt(3) + 1)/
(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3),

y == -1/2*(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3)*(-I*sqrt(3) + 1) + 1/4*(I*sqrt(3) + 1)/
(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3),

y == (1/4*x + 1/4*sqrt(x^2 + 2))^(1/3) - 1/2/(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3)
]
sage: g(x)= sol [2]. rhs ()

sage: g(x)
(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3) - 1/2/(1/4*x + 1/4*sqrt(x^2 + 2))^(1/3)

sage: n( diff (g(x)). substitute (x =2) ,digits =3)
0.207
```

$$f^{-1}(x) = \left(\frac{x}{4} + \frac{\sqrt{x^2+2}}{4}\right)^{\frac{1}{3}} - \frac{1}{2\left(\frac{x}{4} + \frac{\sqrt{x^2+2}}{4}\right)^{\frac{1}{3}}}$$

References:

Granville, William; Joyner, David. **Differential Calculus and Sage**. Dispon[ivel em: https://wdjoyner.files.wordpress.com/2015/04/granville_calc1-sage_2009-08-15.pdf. Acesso em: 04 de ago. 2018.

SageMath, the Sage Mathematics Software System (Version 8.1), The Sage Developers, 2017, Dispon[ivel em: http://www.sagemath.org. Acesso em: 04 de ago. 2018.

Guidorizzi, H. L. . **Um Curso de Cálculo**, Volume 1. Editora LTC, Rio de Janeiro, 2001.

Stewart, James. **Cálculo**, Volume 1 e 2. São Paulo: Cengage Learning, 2016.

Thomas, G.B.; Finney, R. L.; Weir, M. D.; Giordano, F. R. **Cálculo**, Volumes 1 e 2. Editora Pearson Education do Brasil, São Paulo, 2002.

Piskunov. N. Cálculo Diferencial e Integral, Volumes 1 e 2. Editora livraria Lopes da Silva, Porto, 1986.

Le, Tuan A,; Nguyen, Hieu D. **SageMath™ Advice For Calculus**. Disponivel em: http://users.rowan.edu/~nguyen/sage/SageMathAdviceforCalculus.pdf. Acesso em: 07 de ago. 2018.